A BASIC QUESTION ABOUT SELF-NORMALIZING SUBGROUPS

WILLIAM DEMEO

Suppose $H$ and $K$ are incomparable subgroups of a group $G$. If their intersection is self-normalizing, must they both be self-normalizing? In other words, does the following hold?

(1) $N_G(H \cap K) = H \cap K \Rightarrow N_G(H) = H$ and $N_G(K) = K$.

This seemed unlikely, but a proof was not obvious to me, and counter-examples were surprisingly rare. In fact, there are no groups of order $|G| < 96$, and no nonsolvable groups of order $|G| < 360$, containing subgroups which violate (1).

I used GAP [1] to find a counter-example to (1) in the alternating group on six letters. Let $G = A_6$, $H = (C_3 \times C_3) : C_2$, and $K = S_4$.

Then $H$ is strictly contained in its normalizer, $N_G(H) = (C_3 \times C_3) : C_4$, while $H \cap K = S_3$ which is self-normalizing in $A_6$. The Hasse diagram in Figure 2 shows the whole interval above $H \cap K$ in the subgroup lattice Sub[$A_6$].

Incidentally, it is easy to prove that the converse of (1) is false; that is, self-normalizing is not a property that is closed under intersection. (In fact, I believe it’s not hard to prove that for any non-normal subgroup $X$, there exists a conjugate $X^g$ such that $X \cap X^g$ is not self-normalizing.)

Naturally, the next question to ask is whether the following is true:

(2) $N_G(H \cap K) = H \cap K \Rightarrow N_G(H) = H$ or $N_G(K) = K$.

I asked mathoverflow.net for a counter-example (question [3]); specifically, I asked if anyone knew of a group $G$ with two incomparable subgroups $H$ and $K$ such that $N_G(H \cap K) = H \cap K$, yet $N_G(H) \leq H$ and $N_G(K) \leq K$.

Tim Dokchitser quickly replied (answer [2]) with the following counter-example to (2): Let $G \cong S_3 \times S_3 \times S_3$ be the subgroup of $S_9$ generated by $(1,2,3), (1,2), (4,5,6), (4,5), (7,8,9), (7,8)$, and let $U$ be the diagonal subgroup $U = \{(x,x,x) : x \in G\}$, which is generated by $(1,2,3)(4,5,6)(7,8,9)$ and $(1,2)(4,5)(7,8)$. Let

$$H = \langle U, (1,2,3) \rangle \quad \text{and} \quad K = \langle U, (4,5,6) \rangle.$$ 

Then

$$H \cong K \cong (C_3 \times C_3) : C_2 \quad \leq \quad S_3 \times S_3 \cong N_G(H) \cong N_G(K),$$

while $U = H \cap K \cong S_3$ is self-normalizing.

I used GAP to verify these assertions and used XGAP to draw the upper interval $[U,G]$ of the subgroup lattice Sub[$G$] (Figure 1). The GAP and XGAP code appears below.

\textit{Date:} 8 December 2010.
Figure 1. Hasse diagram of the upper interval $[U, G]$ in $\text{Sub}[G]$ drawn by the XGAP program.

Figure 1 shows the upper interval $[U, G]$ in $\text{Sub}[G]$, where $G = S_3^3$ and $U$ is the diagonal subgroup. The four subgroups of index 12 have structure $(C_3 \times C_3) : C_2 = C_3 \wr S_2$. The three subgroups of index 6 have structure $S_3 \times S_3$. The normal subgroup of index 4 has structure $(C_3 \times C_3 \times C_3) : C_2$, and the three maximal normal subgroups have structure $S_3 \times (C_3 \wr S_2)$. 
Figure 2. Hasse diagram of the interval $[S_3, A_6]$ in Sub$[A_6]$ drawn by the XGAP program.
```
GAP code

S3:=SymmetricGroup(3);
G:=DirectProduct(S3,S3,S3);

# Construct U (the diagonal subgroup)
emb1:=Embedding(G,1);; emb2:=Embedding(G,2);; emb3:=Embedding(G,3);;
diag:=List(GeneratorsOfGroup(S3), i->Image(emb1,i)*Image(emb2,i)*Image(emb3,i));
U:=Group(diag,());;

H:=Group(Concatenation(diag, [(1,2,3)]),());
K:=Group(Concatenation(diag, [(4,5,6)]),());

StructureDescription(H);  # (C3 x C3) : C2
StructureDescription(Normalizer(G,H)); # S3 x S3
    # (H is not self-normalizing)

StructureDescription(K);  # (C3 x C3) : C2
StructureDescription(Normalizer(G,K)); # S3 x S3
    # (K is not self-normalizing)

HcapK := Intersection(H,K);;
StructureDescription(HcapK);  # S3
StructureDescription(Normalizer(G,HcapK)); # S3
    # (HcapK is self-normalizing)

# The filter above HcapK is:
filterHcapK := IntermediateSubgroups(G,HcapK);;

# Have GAP store their descriptions.
for k in [1..Size(filterHcapK.subgroups)] do
    StructureDescription(filterHcapK.subgroups[k]);
od;

# There are 11:
# filterHcapK.subgroups := [
#   (C3 x C3):C2, (C3 x C3):C2, (C3 x C3):C2, (C3 x C3):C2,
#   S3 x S3, S3 x S3, S3 x S3,
#   (C3 x C3 x C3):C2, S3 x ((C3 x C3):C2),
#   ((C3 x C3):C2) x S3, ((C3 x C3):C2) x S3 ]

# H and K are filterHcapK.subgroups[3] and [2], resp.

# Using XGAP, we can view this upper interval in the Hasse diagram of Sub[G]:
L := GraphicSubgroupLattice(G);
```
InsertVertex(L, HcapK);
InsertVertex(L, filterHcapK.subgroups[1]);
InsertVertex(L, K);  # filterHcapK.subgroups[2]);
InsertVertex(L, H);  # filterHcapK.subgroups[3]);
InsertVertex(L, filterHcapK.subgroups[4]);

# To keep track of the vertices as they get added, insert them in subsets
# The three S3 x S3:
for k in [5..7] do
    InsertVertex(L, filterHcapK.subgroups[k]);
od;

# The rest:
for k in [8..11] do
    InsertVertex(L, filterHcapK.subgroups[k]);
od;

REFERENCES


   URL: http://mathoverflow.net/questions/48682 (version: 2010-12-08).

   URL: http://mathoverflow.net/questions/48678 (version: 2010-12-09).